



# I PERCHE' DELLA SCIENZA

FABBRI EDITORI

CON 4 PROGRAMMI  
PER COMPUTER

Traduzione italiana a cura di Virginio Sala

Titolo originale: Usborne Introduction to Biology, Chemistry, Physics, Electronics  
© 1983/84/85 Usborne Publishing, London

Per l'edizione in lingua italiana

© 1985 Gruppo Editoriale Fabbri, Bompiani, Sonzogno, Etas S.p.A., Milano  
Ristampa 1991

Finito di stampare nell'anno 1991 presso lo Stabilimento  
Grafico del Gruppo Editoriale Fabbri S.p.A., Milano

ISBN 88-450-2546-2





Jane Chisholm  
David Beeson

# BIOLOGIA

Illustrazioni di  
Sue Stitt, Kuo Kang Chen,  
Graham Round, Chris Shields, Ian Jackson,  
Aziz Khan, Chris Lyon, David Quinn,  
Martin Newton, Rob McGaig

# Un programma di calcolatore per la classificazione di un animale

Se hai un home-computer, o puoi averne uno in prestito, puoi usare questo programma, che ti aiuterà a classificare un animale. Pensa a un animale, poi rispondi alle domande per stabilire a quale phylum appartiene. In alcuni casi troverai anche la sua classe e il suo ordine. Il programma è stato scritto in

modo da funzionare su un Commodore 64 o un VIC 20; le righe che debbono essere modificate per poter eseguire il programma su altri computer sono indicate con alcuni simboli e le sostituzioni sono riportate alla fine del programma. Ciascun simbolo corrisponde a un diverso tipo di computer, per la precisione:

## ▲ ZX Spectrum

## ● Apple

### Come adattare il programma

Questo programma è molto semplice, ma ti può dare un'idea di come i biologi possono utilizzare il computer per classificare più facilmente gli organismi viventi. Il programma è stato scritto in modo che tu possa adattarlo, se vuoi. Puoi espanderlo includendo altre domande e altre risposte per classificare un animale in modo più particolareggiato.

Per formulare un'altra domanda, guarda la riga 3200. Con quale di queste espressioni comincia la tua domanda? Inserisci un numero corrispondente alla fine della riga 3210. Se la domanda comincia con "Ha", inserisci 1, se comincia con "È" inserisci 2, se comincia con "Vive" inserisci 3; inserisci 4 se comincia in un altro modo.

Dopo la riga 1180 aggiungi una nuova riga, 1190, e il resto della tua domanda. Dopo la riga 2090, aggiungi la riga 2100 e la risposta. Se non ci sono altre domande, aggiungi uno 0 alla fine della riga 3100. Se vuoi formulare un'altra domanda, inserisci il suo numero, probabilmente 21. (Così com'è, il programma contiene 19 domande.) Dovrai correggere anche la riga 380. ND sta per il numero delle domande nel programma, e NR per il numero delle risposte.

```
10 GOSUB 380
20 LET N=1
30 GOSUB 320
40 IF N=ND THEN GOSUB 140:GOTO 60
50 IF N>100 THEN GOSUB 210
60 IF N=999 THEN GOTO 120
70 IF N<>0 THEN GOTO 40
80 PRINT:PRINT "VUOI CLASSIFICARE UN ALTRO ANIMALE? "
90 PRINT:GOSUB 300
100 IF A$="S" THEN GOTO 20
110 PRINT "CIAO...":STOP
120 PRINT:PRINT "MI DISPIACE - NON POSSIAMO CLASSIFICARE"
125 PRINT "QUESTO ANIMALE"
130 GOTO 80
140 GOSUB 280
150 PRINT B$(S(N));" ";
160 RESTORE:FOR I=1 TO N:READ M$:NEXT I
170 PRINT M$:PRINT:PRINT:GOSUB 300
180 IF A$="S" THEN LET N=Y(N)
190 IF A$="N" THEN LET N=N(N)
200 RETURN
210 GOSUB 280:GOSUB 360
220 FOR I=1 TO N-100:READ M$:NEXT I
230 PRINT "HAI UN MEMBRO DEL(LA) ";M$
240 LET N=L(N-100):GOSUB 260:RETURN
250 GOSUB 260:RETURN
260 PRINT:PRINT "PREMI RETURN PER CONTINUARE"
270 INPUT A$:RETURN
280 PRINT CHR$(147)
290 RETURN
300 PRINT "RISPONDI S (SI) O N (NO)"
310 INPUT A$:RETURN
320 GOSUB 280:PRINT "PENSA UN ANIMALE":PRINT
330 PRINT "RISPONDI ALLE DOMANDE CHE SEGUONO"
340 PRINT "PER POTERLO CLASSIFICARE"
350 GOSUB 260:RETURN
360 RESTORE:FOR I=1 TO ND:READ M$
370 NEXT I:RETURN
380 LET ND=19:LET NR=19:GOSUB 360
390 DIM Y(ND):DIM N(ND):DIM L(NR):DIM S(ND):DIM B$(4)
400 FOR I=1 TO NR:READ M$:NEXT I
410 FOR I=1 TO ND:READ Y(I):READ N(I):NEXT I
420 FOR I=1 TO NR:READ L(I):NEXT I
430 FOR I=1 TO 4:READ B$(I):NEXT I
440 FOR I=1 TO ND:READ S(I):NEXT I
450 RETURN
```

```
1000 DATA "UNO SCHELETRO INTERNO SOLIDO"
1010 DATA "COSTITUITO DA UNA SOLA CELLULA"
```

```

1020 DATA "IN ACQUA E HA TENTACOLI PUNGENTI"
1030 DATA "UN CORPO MOLLE, DIVISO IN ANELLI O SEGMENTI"
1040 DATA "UNO SCHELETRO ESTERNO"
1050 DATA "DUE PAIA DI ANTENNE E DA 10 A 14 ZAMPE"
1060 DATA "8 ZAMPE E NESSUNA ANTENNA"
1070 DATA "6 ZAMPE E CORPO DIVISO IN 3 SEZIONI"
1080 DATA "CORPO ALLUNGATO E MOLTE ZAMPE"
1090 DATA "CORPO MOLLE RACCHIUSO IN UN GUSCIO"
1100 DATA "NEL MARE E HA PELLE RIVESTITA DI SPINE"
1110 DATA "IN ACQUA, HA SCAGLIE E RESPIRA CON LE BRANCIE"
1120 DATA "PELLE VISCIDA, VIVE SULLA TERRAFERMA MA DEPONE LE UOVA IN ACQUA"
1130 DATA "SULLA TERRAFERMA, HA PELLE A SCAGLIE E DEPONE UOVA CON GUSCIO"
1140 DATA "ALI CON PIUME"
1150 DATA "A SANGUE CALDO E ALLATTA I SUOI PICCOLI"
1160 DATA "DEPONE UOVA"
1170 DATA "UNA TASCA"
1180 DATA "I PICCOLI SI SVILUPPANO NELLA PLACENTA"

2000 DATA "SOTTOPHYLUM VERTEBRATI (PHYLUM CORDATI)"
2010 DATA "PHYLUM PROTOZOI", "PHYLUM CELEENTERATI"
2020 DATA "PHYLUM ANELLIDI", "PHYLUM ARTROPODI"
2030 DATA "PHYLUM CROSTACEI", "CLASSE ARACNIDI"
2040 DATA "CLASSE INSETTI", "CLASSE MIRIAPODI"
2050 DATA "PHYLUM MOLLUSCHI", "PHYLUM ECHINODERMI"
2060 DATA "CLASSE PESCI", "CLASSE ANFIBI"
2070 DATA "CLASSE RETTILI", "CLASSE UCCELLI"
2080 DATA "CLASSE MAMMIFERI", "FAMIGLIA MONOTREMI"
2090 DATA "FAMIGLIA MARSUPIALI", "FAMIGLIA EUTERI"

3000 DATA 101,2,102,3,103,4,104,5,105,10,106,7,107,8,108,9,109,999
3010 DATA 110,11,111,999,112,13,113,14,114,15,115,16,116,999
3020 DATA 117,18,118,19,119,999
3100 DATA 12,0,0,0,0,0,0,0,0,0,0,0,0,0,17,0,0,0
3200 DATA "HA ", "E' ", "VIVE", ""
3210 DATA 1,2,3,1,1,1,1,1,1,1,3,3,1,3,1,2,4,1,1

```

## Per usare altri tipi di computer

Riportiamo qui un elenco di cambiamenti che ti consentiranno di far girare il programma anche su altri calcolatori. I simboli sulla sinistra corrispondono ai diversi tipi di macchina. Queste istruzioni vanno inserite nel programma nei punti opportuni.

```

▲ 150 LET C$=B$(S(N))
▲ 152 IF C$(LEN(C$))=" " THEN LET C$=C$
      (TO LEN(C$)-1): GOTO 152
▲ 155 PRINT C$;" ";
● 280 HOME
▲ 280 CLS:PRINT:PRINT
▲ 390 Cambia DIM B$(4) in DIM B4(4,12)

```



Jane Chisholm  
Mary Johnson

# CHIMICA

Illustrazioni di  
Jeremy Banks, Sue Stitt,  
Chris Lyon, Jeremy Gower, Graham Smith,  
Graham Round, Simon Roulstone, Penny Simon



## Che cosa brucia?

Quando una sostanza libera energia sotto forma di luce e calore, diciamo che sta bruciando. Anche questa è una ossidazione, poiché richiede ossigeno. Senza ossigeno, nulla brucerebbe. Per poter bruciare, la maggior parte delle sostanze deve ricevere energia, sotto forma di calore.



## Riduzione

La riduzione è l'opposto dell'ossidazione. Quando una sostanza cede ossigeno a un'altra sostanza, i chimici dicono che è stata ridotta, mentre l'altra sostanza è stata ossidata. Ossidazione e riduzione, dunque, di solito avvengono nella medesima reazione. I metalli più in alto nella classificazione secondo la loro attività reagiscono fortemente con l'ossigeno e possono togliere ossigeno dai metalli meno reattivi. Questo non vale solo per l'ossigeno. Se si aggiunge un metallo a una soluzione di un composto che contiene un metallo meno reattivo, quest'ultimo verrà "spinto" fuori dalla soluzione dal metallo più reattivo.



Prova a lasciare un chiodo di ferro in una soluzione di solfato di rame. Il rame si raccoglie attorno al chiodo, che diventa rosa. Il ferro ha "spinto" il rame fuori dalla soluzione. Secondo la terminologia dei chimici, è una reazione di scambio.

## I test della fiamma


Quando un metallo brucia, produce una caratteristica fiamma colorata. Questo può servire a stabilire se un dato metallo è presente in un composto sconosciuto. Ti serve un filo non reattivo, per esempio di nicromo, che non interferisca nell'esperimento. Puliscilo in acido cloridrico, poi portalo al calor rosso. Quando la fiamma non ha colore, passalo nel composto sconosciuto e scaldalo di nuovo.



I fuochi artificiali sono fatti con composti che contengono calcio, stronzio (cremisi) e bario.

# Come si identifica una sostanza

Ecco una serie di suggerimenti per aiutarti a svelare l'identità di una sostanza sconosciuta. Puoi provare questi test su una sostanza tipica. Non abbiamo abbastanza spazio per includere tutti i controlli che un chimico può effettuare, ma questi saranno sufficienti per identificare molte fra le sostanze chimiche più comuni. Puoi trovare le spiegazioni di questi test nel volume, ricorrendo all'indice analitico.





## Test delle perline di boro

Scalda un anello di filo di platino \*\*, poi immergilo in acqua distillata. Immergilo in borato di sodio e scaldalo finché ottieni una "perlina" vetrosa. Aggiungi un po' della sostanza sconosciuta e scaldala di nuovo. Sostanze che contengono taluni metalli

producono perline di colore diverso. I colori dipendono dal fatto che siano stati scaldati sulla parte interna o esterna della fiamma.

Prova il test delle perline di boro.

FIAMMA  
INTERNA

FIAMMA  
ESTERNA

CONTIENE

Blu

Cromo

Verde

Cobalto

Rosso/Marrone Turchese

Rame

Verde scuro

Giallo

Ferro


Verde chiaro

Porpora

Manganese

Nero

Rosso/Marrone Nichel



\* Il metallo viene sempre per secondo.

\*\* Per questo esperimento serve un becco bunsen.

# Un programma di calcolatore per l'identificazione di una sostanza

Se hai un home-computer, o puoi averne in prestito uno, puoi usarlo per provare i test per l'identificazione di una sostanza sconosciuta. Il programma è scritto per il Commodore 64 o il VIC 20. Le righe che debbono essere modificate per poter

eseguire il programma su altri computer sono indicate con alcuni simboli e riportate alla fine del programma. Ogni simbolo corrisponde a un diverso computer.

I simboli sono:

▲ ZX Spectrum

● Apple

Il programma è stato stampato in modo da essere più facilmente leggibile (in particolare nelle istruzioni che contengono i dati): nel ribatterlo, non tenere conto di questa impaginazione,

ma batti tutto il contenuto di ciascuna istruzione di seguito, premendo il tasto RETURN solo alla fine dell'istruzione, e non quando va a capo il testo stampato.

```

10 REM * IDENTIFICAZIONE DI SOSTANZE *
20 REM * ----- *
30 GOSUB 930
40 GOSUB 830
50 LET N=1
60 PRINT CHR$(147)
70 REM * CICLO PRINCIPALE *
80 PRINT
90 PRINT Q$(N);
100 LET F=A(N,1)
110 IF F=0 THEN GOSUB 230
120 IF F=1 THEN GOSUB 310
130 IF N<>0 THEN GOTO 80
140 REM * FINE PAGINA *
150 PRINT
160 PRINT "E' TUTTO QUEL CHE POSSIAMO FARE
170 PRINT "VUOI ESEGUIRE IL PROGRAMMA "
180 PRINT "UN'ALTRA VOLTA? (S/N)"
190 INPUT A$
200 IF A$="S" THEN GOTO 40
210 PRINT "VA BENE."
220 STOP
230 REM * STAMPA UNA FRASE *
240 PRINT:PRINT
250 GOSUB 690
260 LET P=A(N,2)
270 IF P=1 THEN GOSUB 360
280 IF P=2 THEN GOSUB 590
290 LET N=A(N,3)
300 RETURN
310 REM * FORMULA UNA DOMANDA *
320 PRINT " ?":PRINT
330 GOSUB 740
340 LET N=A(N,R+1)
350 PRINT CHR$(147):RETURN
360 REM * TEST DELLA FIAMMA *
370 PRINT CHR$(147):PRINT
380 PRINT "TEST DELLA FIAMMA"
390 PRINT

```

```

400 PRINT "GUARDA A PAGINA 75 "
410 PRINT "DEL LIBRO PER VEDERE I"
420 PRINT "PARTICOLARI DEL TEST"
430 PRINT
440 PRINT "HAI UN RISULTATO?"
450 GOSUB 710
460 IF R=2 THEN GOSUB 480
470 RETURN
480 REM * TEST DELLA PERLINA DI BORO *
490 PRINT CHR$(147):PRINT
500 PRINT "TEST DELLA PERLINA DI BORO"
510 PRINT
520 PRINT "GUARDA ALLE PAGINE 86-87 "
530 PRINT "DEL LIBRO PER VEDERE "
540 PRINT "I PARTICOLARI DI QUESTO TEST"
550 PRINT
560 GOSUB 690
570 PRINT CHR$(147)
580 RETURN
590 REM * TEST DEGLI ACIDI *
600 PRINT CHR$(147):PRINT
610 PRINT "TEST DEI RADICALI ACIDI":PRINT
620 PRINT
630 PRINT "GUARDA ALLE PAGINE 86-87 "
640 PRINT "DEL LIBRO PER VEDERE "
650 PRINT "I PARTICOLARI DI QUESTO TEST"
660 PRINT
670 GOSUB 690
680 RETURN
690 REM * VIENE PREMUTO UN TASTO *
700 PRINT "PREMI LO SPAZIO PER CONTINUARE"
710 GET A$
720 IF A$<>" " THEN GOTO 710
730 RETURN
740 REM * RICEVI LA RISPOSTA S/N *
750 PRINT
760 LET R=0
770 PRINT "RISPONDI S (SI) O N (NO)"
780 INPUT A$

```

```

790 IF A$="S" THEN LET R=1
800 IF A$="N" THEN LET R=2
810 IF R=0 THEN GOTO 780
820 RETURN
830 REM * PAGINA INTRODUTTIVA *
●▲ 840 PRINT CHR$(147):PRINT
850 PRINT "IDENTIFICAZIONE DI SOSTANZE"
860 PRINT "-----"
870 PRINT
880 PRINT "PROVA QUESTI TEST SU "
890 PRINT "CIASCUNA SOSTANZA."
900 PRINT
910 GOSUB 690
●▲ 920 PRINT CHR$(147):RETURN
930 REM * LEGGI I DATI *
▲ 940 DIM Q$(54),A(54,3)
950 LET K=1
960 READ A$
970 IF A$="FINE DEI DATI" THEN RETURN
980 LET Q$(K)=A$
990 FOR I=1 TO 3
1000 READ A(K,I)
1010 NEXT I
1020 LET K=K+1
1030 GOTO 960
1040 REM * I DATI *
1050 DATA "CONDUCE CALORE O ELETTRICITA'",
1,2,5,"E' UN METALLO. E' MAGNETICO",
1,3,4
1060 DATA "E', O CONTIENE FERRO",0,0,0,
"PROVA I TEST DELLA FIAMMA",0,1,0
1070 DATA "FONDE FACILMENTE",1,6,17, "E'
UN COMPOSTO O UN ELEMENTO COVALENTE",
0,0,7
1080 DATA "E' UNA POLVERE GIALLA",1,10,8,
"O UN SOLIDO ROSSO/GIALLO",1,11,9
1090 DATA "O UN SOLIDO VIOLA/NERO",1,12,13,
"E' PROBABILMENTE ZOLFO",0,0,0
1100 DATA "PROBABILMENTE E' FOSFOROSO",0,
0,0,"E' PROBABILMENTE IODIO",0,0,0
1110 DATA "POTREBBE ESSERE UN COMPOSTO
ORGANICO",0,0,14,"SCALDALO. VIENE
LIBERATA ANIDRIDE CARBONICA",1,16,15
1120 DATA "NON E' ORGANICO",0,0,0,"E'
ORGANICO E CONTIENE CARBONIO",0,0,0
1130 DATA "PROBABILMENTE E' UN COMPOSTO
IONICO",0,0,18,"SI SCIOLGIE",1,24,19
1140 DATA "E' UNA POLVERE NERA",1,20,23,
"SCALDALO. VIENE LIBERATA ANIDRIDE
CARBONICA",1,21,22
1150 DATA "E' CARBONIO",0,0,0,"E' OSSIDO
DI RAME",0,0,0
1160 DATA "E' UN SALE INSOLUBILE",0,0,29,
"TROVA IL SUO PH",0,0,25
1170 DATA "E' UN ACIDO",1,42,26,"E' UN
ALCALI",1,33,27

```

```

1180 DATA "E' NEUTRO",1,28,25,"E' UN SALE
NEUTRO",0,0,29
1190 DATA "TROVA IL METALLO CONTENUTO",
0,1,30,"TROVA L'ACIDO CONTENUTO",
0,2,31
1200 DATA "COMBINA IL METALLO E IL RADICALE
ACIDO",0,0,32,"PER TROVARE IL NOME DEL
COMPOSTO",0,0,0
1210 DATA "POTREBBE ESSERE UN OSSIDO O...",
0,0,34,"UN IDROSSIDO DI UN METALLO",
0,0,35
1220 DATA "HA ODORE DI AMMONIACA",1,54,36,
"FAI IL TEST DEL RADICALE ACIDO",0,2,37
1230 DATA "HAI UN RISULTATO",1,39,38,"E'
PROBABILMENTE UN OSSIDO O UN IDROSSIDO",
0,0,39
1240 DATA "TROVA IL METALLO CONTENUTO",0,1,40
"COMBINALO CON IL RADICALE ACIDO",0,0,41
1250 DATA "OPPURE OSSIDO O IDROSSIDO PER
COMPOSTO",0,0,0,"AGGIUNGI ZINCO. VIENE
LIBERATO IDROGENO",1,44,43
1260 DATA "E' UN SALE ACIDO", 0,0,29,"E' UN
ACIDO", 0,0,45
1270 DATA "FAI IL TEST DEL RADICALE ACIDO",
0,2,46,"E' IL RADICALE SOLFATO",1,49,47
1280 DATA "DATA "O CLORURO",1,50,48,"O
CARBONATO",1,51,52
1290 DATA "E' ACIDO SOLFORICO",0,0,0,"E'
ACIDO CLORIDRICO",0,0,0
1300 DATA "E' ACIDO CARBONICO",0,0,0,
"POTREBBE ESSERE UN NITRATO O...",0,0,53
1310 DATA "BROMURO O IODURO",0,0,0,"E' UN
COMPOSTO DI AMMONIO",0,0,0
1320 DATA "FINE DEI DATI"

```

## Per usare altri tipi di computer

Riportiamo qui un elenco di cambiamenti che ti consentiranno di far girare il programma anche su altri calcolatori. I simboli sulla sinistra corrispondono ai diversi tipi di macchina. Queste istruzioni vanno inserite nel programma nei punti opportuni.

▲ 710 LET A\$=INKEY\$

▲ 940 DIM Q\$(54,38):DIM A(54,3)


▲ 60,350,370,490,570,600,840,920: sostituisci il PRINT CHR\$(147) con CLS

● 60,350,370,490,570,600,840,920: sostituisci il PRINT CHR\$(147) con HOME

● 705 A\$=""

● 710 IF PEEK-16384>127 THEN GET A\$





Amanda Kent  
Alan Ward

# FISICA

Illustrazioni di  
Sue Stitt, Chris Lyon, Jeremy Gower,  
Simon Roulstone, Mick Gillah

## Un programma di calcolatore per l'elettricità domestica

Questo programma stabilisce quanta elettricità consumano oggetti come il televisore e il forno.

Puoi anche calcolare a quanto potrà approssimativamente ammontare la prossima bolletta trimestrale dell'energia elettrica.

Se possiedi, o puoi usare, un VIC 20 o

un Commodore 64, il programma può essere eseguito così com'è; le righe che debbono essere modificate per poterlo eseguire su altri computer, sono indicate con alcuni simboli e le sostituzioni sono riportate alla fine del programma stesso. Ciascun simbolo corrisponde a un diverso tipo di computer, per la precisione:

## ▲ ZX Spectrum

Prima di poter provare a calcolare la prossima bolletta, devi guardare su una bolletta recente il prezzo unitario dell'elettricità.

Le differenze fra le varie macchine sono troppo grandi per poter dare istruzioni generali per la grafica, ma riportiamo un esempio di sottoprogramma grafico per

● Apple

disegnare una lampadina, che funziona su uno Spectrum. Il sottoprogramma va aggiunto alla fine del programma, inserendo una nuova istruzione che lo chiami al momento opportuno: 1675 GOSUB 3000. Puoi magari provare a scrivere altri sottoprogrammi grafici per gli altri elettrodomestici.

```

10 REM INIZIALIZZA
20 LET N=10: REM NUMERO DI
    ELETTRODOMESTICI
30 DIM U(N): REM UNITA' USATE
40 DIM A$(N): REM NOMI
50 LET TU=0: REM POTENZA USATA
60 LET UP=2.5: REM PREZZO UNITARIO
70 LET A$(1) = "CUCINA"
80 LET A$(2) = "RISCALDATORE A
    ' IMMERSIONE"
90 LET A$(3) = "TERMOVENTILATORE"
100 LET A$(4) = "RADIATORE"
110 LET A$(5) = "LAMPADINA"
120 LET A$(6) = "LAVATRICE"
130 LET A$(7) = "TELEVISORE"
140 LET A$(8) = "RADIO"
150 LET A$(9) = "STUFETTA A CONVEZIONE"
160 LET A$(10) = "STEREO HI-FI"
170 REM * STAMPA PAGINA INTRODUTTIVA *
180 PRINT CHR$(147)
190 PRINT
200 PRINT
210 PRINT "CALCOLO DELLA BOLLETTA"
220 PRINT "    DELL'ELETTRICITA'  "
230 PRINT "===== "
240 PRINT
250 PRINT "CENTRALE"
260 PRINT "DI POTENZA>>>>>>>>>>"
270 PRINT "                                TRASFOR-"
280 PRINT "                                MATORE"
290 FOR I = 1 TO 4
300 PRINT "                                V"

```

```

310 NEXT I
320 PRINT "                                ABITAZIONE"
330 PRINT
340 PRINT "PREMI SPAZIO PER CONTINUARE"
350 GOSUB 810
360 REM * PAGINA DEL MENU' PRINCIPALE *
370 PRINT CHR$(147)
380 PRINT "SCEGLI L'ELETTRODOMESTICO"
390 PRINT "DA INSERIRE ULTERIORMENTE"
400 PRINT "OPPURE PREMI 0 (ZERO)"
410 PRINT "PER CALCOLARE LA BOLLETTA"
420 PRINT
430 PRINT "                                UNITA'"
440 FOR I = 1 TO N
450 IF U(I)>0 THEN PRINT ;I;" ";A$(I);
    TAB(19);U(I)
460 IF U(I)=0 THEN PRINT ;I;" ";A$(I)
470 NEXT I
480 PRINT
490 PRINT "BATTI UN NUMERO, QUINDI"
500 PRINT "PREMI IL TASTO RETURN ";
510 INPUT C
520 IF C<0 OR C>N THEN GOTO 360
530 IF C=0 THEN GOTO 580
540 PRINT CHR$(147)
550 PRINT
560 ON C GOSUB 1060,1280,1330,1530,
    1650,1700,1900,2060,2110,2160
570 GOTO 360
580 REM PAGINA FINALE
590 PRINT CHR$(147)
600 FOR W=1 TO N

```

```

610 LET TU=TU+U(W)
620 NEXT W
630 PRINT
640 PRINT "      STIMA      "
650 PRINT " DELLA BOLLETTA "
655 PRINT "DELL'ELETTRICITA'"
660 PRINT "===== "
670 PRINT " (PER 3 MESI)  "
680 PRINT
690 PRINT "UNITA' USATE: "
700 PRINT ;TU;" KILOWATTORA"
710 PRINT
720 PRINT "PREZZO UNITARIO: ";UP;
    "LIRE"
730 LET TC=(INT(UP*TU))/100
740 PRINT
750 PRINT
760 PRINT "TOTALE DOVUTO: ";TC
770 PRINT
780 PRINT "PREMI SPAZIO PER
    RICOMINCIARE"
790 GOSUB 810
800 RUN
810 GET I$
820 IF I$<>" " THEN GOTO 810
830 RETURN
840 REM * PAGINA PER IMMISSIONE DATI *
850 PRINT CHR$(147)
860 PRINT
870 PRINT N$
880 PRINT
890 PRINT ;P*1000;" WATT"
900 FOR I = 1 TO 7
910 PRINT
920 NEXT I
930 PRINT "PER QUANTO TEMPO VIENE
    USATO"
940 PRINT "QUESTO ELETTRODOMESTICO, IN
    MEDIA"
950 PRINT "OGNI SETTIMANA (IN ORE)"
960 PRINT "BATTI IL NUMERO, POI"
970 PRINT "PREMI IL TASTO RETURN";
980 INPUT T
990 LET U(C)=U(C)+P*T*13
1000 RETURN
1010 REM * SPOSTA IN GIU' DI 5 RIGHE *
1020 FOR X = 1 TO 5
1030 PRINT
1040 NEXT X
1050 RETURN
1060 REM * CUCINA *
1070 PRINT A$(C)
1080 GOSUB 1010
1090 PRINT "PREMI 1) PER IL FORNELLO"
1100 PRINT "      2) PER IL FORNO"
1110 PRINT "      3) PER IL GRILL"
1120 PRINT
1130 INPUT I
1140 IF I<1 OR I>3 THEN GOTO 1130
1150 ON I GOTO 1160,1200,1240

1160 LET N$="FORNELLO DELLA CUCINA"
1170 LET P=1
1180 GOSUB 840
1190 RETURN
1200 LET N$="FORNO DELLA CUCINA"
1210 LET P=3
1220 GOSUB 840
1230 RETURN
1240 LET N$="GRILL DELLA CUCINA"
1250 LET P=1.5
1260 GOSUB 840
1270 RETURN
1280 REM * RISCALDATORE A IMMERSIONE *
1290 LET N$=A$(C)
1300 LET P=3.5
1310 GOSUB 840
1320 RETURN
1330 REM * TERMOVENTILATORE *
1340 LET N$="TERMOVENTILATORE"
1350 PRINT N$
1360 GOSUB 1010
1370 PRINT "E'  1) AL MASSIMO"
1380 PRINT "      2) A META'"
1390 PRINT "      3) SULL'ARIA FREDDA"
1400 INPUT I
1410 IF I<1 OR I>3 THEN GOTO 1400
1420 ON I GOTO 1430,1460,1490
1430 LET N$=N$+" (AL MASSIMO)"
1440 LET P=3
1450 GOTO 1510
1460 LET N$=N$+" (A META')"
1470 LET P=1.5
1480 GOTO 1510
1490 LET N$=N$+" (ARIA FREDDA)"
1500 LET P=0.3
1510 GOSUB 840
1520 RETURN
1530 REM * RADIATORE *
1540 LET N$="RADIATORE"
1550 PRINT N$
1560 GOSUB 1010
1570 PRINT "STAI USANDO "
1580 PRINT "1, 2 O 3 ELEMENTI".
1590 INPUT I
1600 IF I<1 OR I>3 THEN GOTO 1590
1610 LET N$=N$+" (" +STR$(I)+
    "ELEMENTI)"
1620 LET P=I
1630 GOSUB 840
1640 RETURN
1650 REM * LAMPADINA *
1660 LET N$=A$(C)
1670 LET P=0.1
1680 GOSUB 840
1690 RETURN
1700 REM * LAVATRICE *
1710 LET N$="LAVATRICE"
1720 PRINT N$
1730 GOSUB 1010
1740 PRINT "STA  1) LAVANDO"

```



```

1750 PRINT "      2) CENTRIFUGANDO"
1760 PRINT "      3) RISCALDANDO"
1770 INPUT I
1780 IF I<1 OR I>3 THEN GOTO 1770
▲ 1790 ON I GOTO 1800,1830,1860
1800 LET N$=N$+" (STA LAVANDO)"
1810 LET P=0.8
1820 GOTO 1880
1830 LET N$=N$+" (STA CENTRIFUGANDO)"
1840 LET P=0.8
1850 GOTO 1880
1860 LET N$=N$+" (STA RISCALDANDO)"
1870 LET P=3
1880 GOSUB 840
1890 RETURN
1900 REM * TELEVISIONE *
1910 LET N$="TELEVISIONE"
1920 PRINT N$
1930 GOSUB 1010
1940 PRINT "E"      1) A COLORI"
1950 PRINT " OPPURE 2) BIANCO E NERO"
1960 INPUT I
1970 IF I<1 OR I>2 THEN GOTO 1960

```

```

1980 IF I=2 THEN GOTO 2020
1990 LET N$=N$+" (A COLORI)"
2000 LET P=0.4
2010 GOTO 2040
2020 LET N$=N$+" (BIANCO E NERO)"
2030 LET P=0.3
2040 GOSUB 840
2050 RETURN
2060 REM * RADIO *
2070 LET N$=A$(C)
2080 LET P=0.05
2090 GOSUB 840
2100 RETURN
2110 REM * STUFETTA A CONVEZIONE *
2120 LET N$=A$(C)
2130 LET P=3
2140 GOSUB 840
2150 RETURN
2160 REM * STEREO HI-FI *
2170 LET N$=A$(C)
2180 LET P=0.15
2190 GOSUB 840
2200 RETURN

```

## Grafica per la lampadina

Questo è un sottoprogramma grafico per disegnare una lampadina. Funzionerà solo su uno Spectrum, e va aggiunto alla fine del programma. Per la chiamata, si deve inserire nel programma una nuova riga: 1675 GOSUB 3000.

```


3000 REM GRAFICA PER LA LAMPADINA
3010 CLS : PLOT 175,40: DRAW 0,32:
      DRAW -8,32,.7: DRAW 48,0,-4.9:
      draw -8,-32,.7: DRAW 0,-32
3020 PLOT 184,40: DRAW -8,88,.2
3030 PLOT 199,40: DRAW 8,88,-.2
3040 PRINT AT 5,22: INK 6: BRIGHT 1: "****"
3050 RETURN

```

## Per usare altri tipi di computer

Riportiamo qui un elenco di cambiamenti che ti consentiranno di far girare il programma anche su altri calcolatori. I simboli sulla sinistra corrispondono ai diversi tipi di macchina. Queste istruzioni vanno inserite nel programma nei punti opportuni.

- ▲ 40 DIM A\$(10,16)
- ▲ 560 GOSUB 1060\*(C=1)+1280\*(C=2)+1330\*(C=3)+1530\*(C=4)+1650\*(C=5)+1700\*(C=6)+1900\*(C=7)+2060\*(C=8)+2110\*(C=9)+2160\*(C=10)
- 810 LET I\$=""
- 812 IF PEEK(-16384)>127 THEN GET I\$
- ▲ 810 LET I\$=INKEY\$
- ▲ 1150 GOTO 1160\*(I=1)+1200\*(I=2)+1240\*(I=3)
- ▲ 1420 GOTO 1430\*(I=1)+1460\*(I=2)+1490\*(I=3)
- ▲ 1790 GOTO 1800\*(I=1)+1830\*(I=2)+1860\*(I=3)



Pam Beasant

# ELETRONICA

Illustrazioni di  
Martin Newton, Jane Andrews,  
Gerry Browne, Chris Lyon, Simon Roulstone,  
Jeremy Banks

# Un programma di calcolatore per il codice di colore

Questo è un programma di calcolatore che può aiutarti a identificare i resistori per costruire circuiti. Ti dà il valore di un resistore quando inserisci i colori delle strisce, oppure i colori delle strisce quando inserisci il valore del resistore. Il programma ti dà anche un elenco

completo del codice di colore. Il programma funziona su un Commodore 64 o un Vic 20; le righe che debbono essere modificate per poter eseguire il programma su altri computer sono indicate con alcuni simboli e riportate alla fine del programma. I simboli sono:

## ▲ ZX Spectrum

## ● Apple

### Come usare il programma

Digita il programma, poi premi RETURN (o il tasto equivalente sul tuo calcolatore). Ti si presenteranno tre scelte, che costituiscono il "menù" del programma.

Premi A se vuoi semplicemente l'elenco dei

colori e i numeri corrispondenti.

Premi B e poi inserisci il valore del resistore, per sapere di quale colore debbono essere le sue strisce.

Premi C e poi batti i colori delle strisce, se vuoi sapere il valore del resistore in ohm.

```
10 GOSUB 490
20 GOSUB 470
30 PRINT "AIUTO PER IL CODICE DI COLORE"
40 PRINT:PRINT "A. ELENCO DEI CODICI DI COLORE"
50 PRINT "B. DAL VALORE AI COLORI"
60 PRINT "C. DAI COLORI AL VALORE"
70 PRINT:PRINT "BATTI A, B OPPURE C"
80 INPUT I$
90 IF I$="A" THEN GOSUB 130
100 IF I$="B" THEN GOSUB 180
110 IF I$="C" THEN GOSUB 320
120 GOTO 20
130 GOSUB 470
140 PRINT "I CODICI DI COLORE SONO :":PRINT
150 FOR I=1 TO 10
160 PRINT TAB(2);C$(I);TAB(14);I-1
170 NEXT I:GOSUB 450:RETURN
180 GOSUB 470
190 PRINT "QUAL E' IL VALORE":INPUT V
200 PRINT "QUAL E' IL MOLTIPLICATORE":PRINT "M=MEGAOHM"
210 PRINT "K=KILOOHM":PRINT "O=OHM"
220 INPUT M$
230 IF M$="M" THEN LET V=V*1E6
240 IF M$="K" THEN LET V=V*1E3
250 LET NZ=0
260 IF V>=100 THEN LET V=V/10:LET NZ=NZ+1:GOTO 260
270 LET V=INT(V):PRINT:PRINT "IL CODICE DI COLORE E'"
280 PRINT C$(INT(V/10)+1);" ";
290 PRINT C$((V/10-INT(V/10))*10+1.1);" ";
300 PRINT C$(NZ+1):GOSUB 450
310 RETURN
320 GOSUB 470
330 FOR I=1 TO 3
340 PRINT "INSERISCI IL COLORE ";I:INPUT I$
350 LET F=0:FOR J=1 TO 10
▲ 360 IF I$=C$(J) THEN LET F=1:LET C(I)=J-1
370 NEXT J:IF F=0 THEN PRINT "CONTROLLO ORTOGRAFIA"
:GOTO 340
```

```
380 NEXT I
390 LET V=(C(2)+10*C(1))*10^C(3):LET M$=""
400 IF V>=1E6 THEN LET V=V/1E6:LET M$="MEGA"
410 IF V>=1E3 AND M$<>"MEGA" THEN LET V=V/1E3
:LET M$="KILO"
420 PRINT:PRINT "IL VALORE E' ";V;" ";M$;"OHM"
430 GOSUB 450
440 RETURN
450 PRINT:PRINT "PREMI RETURN PER CONTINUARE"
460 INPUT X$:RETURN
●▲ 470 PRINT CHR$(147)
480 PRINT:RETURN
▲ 490 DIM C$(10):DIM C(3)
500 FOR I=1 TO 10:READ C$(I):NEXT I
510 RETURN
520 DATA "NERO","MARRONE","ROSSO"
530 DATA "ARANCIO","GIALLO","VERDE"
540 DATA "BLU","VIOLA","GRIGIO"
550 DATA "BIANCO"
```

### Per usare altri tipi di computer

Riportiamo qui un elenco di cambiamenti che ti consentiranno di far girare il programma anche su altri calcolatori. I simboli sulla sinistra corrispondono ai diversi tipi di macchina. Queste istruzioni vanno inserite nel programma nei punti opportuni.

```
▲360 IF I$=C$(J) (TO LEN(I$)) THEN LET F=1
:LET C(I)=J-1
▲470 CLS
▲490 DIM C$(10,6):DIM C(3)
●470 HOME
```





Perché le rose hanno le spine e i ricci gli aculei? Perché una nave galleggia? Che cosa c'è all'interno di un atomo? Che cos'è un "chip"? A questi e a tanti altri "perché" della scienza il volume risponde in maniera esauriente e divertente proponendo facili esperimenti, giochi e quiz. Inoltre, per rendere più interessanti e attuali gli argomenti trattati, sono stati inseriti semplici programmi per home-computer, che attraverso il linguaggio dell'informatica stimolano nel lettore l'interesse per le discipline scientifiche.



ISBN 88-450-2546-2



9 788845 025464

